



MSC Patran

Patran Sessionfile Programmierung - Gruppennamen umbenennen

Oft passiert es, dass Gruppennamen in der Patran Database entstanden sind, die Leerzeichen im Namen enthalten. Das kann zum Beispiel auch beim Import einer Geometriedatei passieren. Dann mag es vorkommen, daß man ein unerwartetes Verhalten bei der Nutzung dieser Patran Datenbank feststellt.

Ehe man diese Datenbank löscht und neu beginnt, lohnt es sich einen näheren Blick auf die Patran Datenbank zu werfen. Wenn ein Neuimport in eine frische Patran Datenbank nicht hilft, so sollte man einen Blick auf die Gruppennamen werfen. Findet man dort Leerzeichen in den Namen, so kann man mit Hilfe des folgenden Patran Sessionfiles eine Korrektur vornehmen.

So geht's:

Um ein vollständiges Beispiel zu liefern, soll als erstes eine Patran Datenbank (PDB) erzeugt werden, welche Gruppennamen mit Leerzeichen enthält.

Nach dem Erzeugen einer neuen leeren Patran Datenbank lässt man das folgende Patran Sessionfile ablaufen, welches eine einfache Fläche erzeugt, diese vernetzt und Patran Gruppen erzeugt


```
$ Sessionfile für die Erzeugung eines Basismodells mit Gruppen,  
$ welche Leerzeichen im Namen enthalten  
$  
$ Erzeugt eine Fläche  
STRING ids[VIRTUAL]  
asm_const_patch_xyz("1", "<1 1 0>", "[0 0 0]", "Coord 0", ids)  
$  
$ Vernetzen der Fläche  
ui_exec_function( "mesh_seed_display_mgr", "init" )  
INTEGER fem_create_mesh_surfa_num_nodes  
INTEGER fem_create_mesh_surfa_num_elems  
STRING fem_create_mesh_s_nodes_created[VIRTUAL]  
STRING fem_create_mesh_s_elems_created[VIRTUAL]  
fem_create_mesh_surf_4( "IsoMesh", 49152, "Surface 1", 1, ["0.2"], "Quad4", @  
"#", "#", "Coord 0", "Coord 0", fem_create_mesh_surfa_num_nodes, @  
fem_create_mesh_surfa_num_elems, fem_create_mesh_s_nodes_created, @  
fem_create_mesh_s_elems_created )  
$  
$ Erzeugen von Gruppennamen mit Leerzeichen  
sys_poll_option( 2 )  
ga_group_create( "node list 1" )  
ga_group_entity_add( "node list 1", "Node 1:10" )  
ga_group_create( "node list 2" )  
ga_group_entity_add( "node list 2", "Node 11:12" )  
ga_group_create( "element list 3" )  
ga_group_entity_add( "element list 3", "Element 1:#" )  
sys_poll_option( 0 )  
repaint_graphics( )
```



MSC Patran

Patran Sessionfile Programmierung

Schaut man sich die Gruppenliste in Patran an, so findet man folgendes Bild vor:



Deutlich sind hier die Gruppennamen mit enthaltenen Leerzeichen zu erkennen.



MSC Patran

Patran Sessionfile Programmierung

Nun wird das Patran Command Language Skript (in diesem Fall ein einfaches Patran Sessionfile) ausgeführt.

```
$ Sessionfile für eine Umbenennung von Gruppennamen mit Leerzeichen
$ Ersetzen des Leerzeichens durch einen Unterstrich <_>
$ Variablendeklaration
INTEGER num_groups /* Anzahl der Gruppen in der Patran Database */
INTEGER i, ii /* Indexvariablen */
INTEGER pos /* Position, an der ein Leerzeichen gefunden wird */
STRING group_list[32] (VIRTUAL) /* Liste der in der PDB enthaltenen Gruppen */
STRING current_grname[32] /* Aktueller Gruppenname */
STRING new_grname[32] /* Neu zu vergebender Gruppenname */
$
$ Gibt die Anzahl der in der Patran Datenbank enthaltenen Gruppen zurück
ga_group_ngroups_get(num_groups)
$
$ Stellt den notwendigen Speicherbereich für die Gruppennamen bereit
SYS_ALLOCATE_ARRAY(group_list, 1, num_groups)
$
$ Gibt die in der Patran Datenbank enthaltenen Gruppen zurück
ga_group_groups_get(group_list)
dump group_list
$
$ Schleife über alle gefundenen Gruppen
FOR (i = 1 TO num_groups) LOOP_PDB_GROUPS
$   Aktueller Gruppenname
   current_grname = group_list(i)
$
$ Zuweisen des aktuellen Gruppennamens zu der Variable des neuen Gruppennamens
   new_grname = current_grname
$
$ Schleife über die einzelnen Zeichen des Gruppennamens
   FOR (ii = 1 TO 32) LOOP_ZEICHENKETTE
$     Suche, ob ein Leerzeichen in der aktuellen Zeichenkette des Gruppennamens
$     gefunden wird
     pos = str_index(new_grname, " ")
     IF (pos != 0) THEN
$
```



MSC Patran

Patran Sessionfile Programmierung


```
$      Ein Leerzeichen wurde gefunden, aber nun wird überprüft, ob davor oder
$      danach bereits ein Unterstrich vorkommt
$      Falls ja, dann wird das Leerzeichen einfach entfernt
      IF (str_substr(new_grname, pos-1, 1) == "_") THEN
          new_grname = str_substr(new_grname, 1, pos-1) // @
                      str_substr(new_grname, pos+1, 32)
          CONTINUE LOOP_ZEICHENKETTE
      ENDIF
      IF (str_substr(new_grname, pos+1, 1) == "_") THEN
          new_grname = str_substr(new_grname, 1, pos-1) // @
                      str_substr(new_grname, pos+1, 32)
          CONTINUE LOOP_ZEICHENKETTE
      ENDIF
$
$      Falls nein, dann wird das Leerzeichen durch einen Unterstrich ersetzt
      str_assign(new_grname, pos, 1, "_")
$
$      Wird kein Leerzeichen mehr gefunden, wird die Schleife abgebrochen
      ELSE
          BREAK LOOP_ZEICHENKETTE
      ENDIF
      END FOR /* Ende LOOP_ZEICHENKETTE */
$
$      Umbenennen der aktuellen Gruppe mit dem neuen Gruppennamen
      ga_group_rename(current_grname, new_grname)
$
      END FOR /* Ende LOOP_PDB_GROUPS */
$
$      Gibt die in der Patran Datenbank enthaltenen Gruppen zurück
      ga_group_groups_get(group_list)
      dump group_list
```



MSC Patran

Patran Sessionfile Programmierung

Nach Ausführung dieses Sessionfiles ergibt sich folgende Liste von Patran Gruppennamen.



Es sind nun keine Leerzeichen in den Namen mehr enthalten.



MSC Patran

Patran Sessionfile Programmierung

Auch in der aktuellen Patran Session findet man die Liste der alten und neuen Namen wieder.

Alte Namen:

```
dump group_list
$# STRING group_list[32](VIRTUAL) = ["default_group", "element list 3", @
$#   "node list 1", "node list 2"]
$
```

Neue Namen:

```
dump group_list
$# STRING group_list[32](VIRTUAL) = ["default_group", "element_list_3", @
$#   "node_list_1", "node_list_2"]
```